Query By Form

By Tom Wickerath

Query by Form (QBF, a.k.a. Index Search form) is a technique that allows users to easily find records. Many times, people using Access applications have never experienced the true power of QBF. In this session, we will create a QBF form to help locate Products in the Northwind sample database that ships with Access 2003. I choose this table because it includes two lookup tables, Suppliers and Categories. While the Products table includes only (77) records, one can easily imagine a similar table with thousands of records. The relationships view is shown below:


A well-designed search form allows the developer to minimize network traffic in a multi-user Access application. This generally results in a more satisfying user experience, because your application seems "snappy" to the end-user. The QBF form we will create includes the following characteristics:


· An unbound (no recordsource) form with unbound (no control source) controls that allow the user to enter search criteria.

· Criteria entered into different controls is "ANDed" together, reducing the number of hits. Multi-select list boxes will be provided to allow the user to select multiple criteria within a control. Selections within a control are "ORed" together, increasing the number of hits.

· In this example, to help keep things simple, the SELECT, JOIN and ORDER BY parts of a Structured Query Language (SQL) statement, used as the recordsource for the subform, will be fixed; only the WHERE clause will be created "on-the-fly".

· Search results are displayed in a read-only subform. The subform displays only those fields required to allow the user to identify the record that they wish to open for editing.

· The user is allowed to open one record at a time for editing.

· The QBF form is based in large part on a sample database from chapter 8 of Access 2000 Power Programming, written by F. Scott Barker (SAMS Publishing).

The Products form in the Northwind sample database (Access 2003 version) appears as shown below. Let's say that we want to allow a user to search by the fields shown in green highlight: Product Name, Supplier, Category, Unit Price, Reorder Level, and Discontinued.

Step 1: Prepare the main QBF form

1. Start by making a copy of the Products form.

2. Delete controls on the form that we will not use for searching. 

3. Deselect View | Form Header / Footer and confirm your choice at the resulting dialog.

4. On the Format tab of the Properties dialog, change the caption, set the allowed views to Form View only, with all other views set to No. Remove the navigation buttons (since this will be an unbound form), and set Dividing Lines to No.
Note: If your starting form includes a Record Selector, remove it as well.

5. Using the Format menu, change the Autoformat to Standard.

6. Set the Has Module property on the Other tab to No to remove the existing class module associated with the Products form.

Step 2: Prepare the QBF subform

1. Make a copy of your new main form used for entering search criteria. Name it frmProductsQBFSubform.

2. Change the Default view to Datasheet. Set the allowed views appropriately (ie. Datasheet only). 

3. Set the Allow Filters, Allow Edits, Allow Deletions, Allow Additions and Data Entry properties to No.

4. Change the Supplier and Category controls from combo boxes to text boxes.

5. Rearrange the controls in design view, from top to bottom, in the order that you want to see them in the subform. Click on View | Tab Order... and select Auto Order. The tab order determines the left-to-right layout order in datasheet view.

Notes: 

I prefer to remove the colons from the labels. 

You can leave the Supplier and Category controls as combo boxes, but this can lead to user confusion, since the user will not be able to change the values if you set the form as read-only.

6. Open the form in preview mode. Resize the controls and the form appropriately.

7. Create a query for the subform that will allow you to display data for the fields in your subform. Include the primary key, ProductID, since it will be used later to identify which record to open for editing. Save the query as "qryQBF":

SELECT ProductID, ProductName, CompanyName AS Supplier, 

CategoryName AS Category, UnitPrice, ReorderLevel, Discontinued

FROM Categories 

RIGHT JOIN (Suppliers RIGHT JOIN Products ON Suppliers.SupplierID = Products.SupplierID) 

ON Categories.CategoryID = Products.CategoryID;

Note: I changed the default Inner Joins to Outer Joins, so that our query will pick up all products, even if a Supplier or Category value is missing.

8. Set the recordsource for the subform such that no records will be returned. This will cause the subfrom to open without displaying any records:
 


SELECT * FROM qryQBF WHERE False

Note: I like to add a label to the subform with this note in red font, as a reminder to myself and any other developers who may follow. Set the visible property for this label to No.

The Products subform with a recordset that returns zero records:

Step 3: Add your new subform to the main QBF form
 

1. Remove the control source from the controls on the main QBF search form.


2. Set the Control Box, Min Max Buttons, and Close Button to No, None and No, respectively. Add a command button to allow the user to close this form.
 

3. Convert the Supplier and Category combo boxes on the main form to list boxes if you want to allow multiple selections. Change the Multiselect property for these two list boxes from None to either Simple or Extended.
 

4. Add a frame for Product Status with three option buttons: Active, Discontinued and Both. We will use this new frame to replace the check box that we had earlier.
 

5. Add a frame with two option buttons: Auto Requery and No Auto Requery.
Add two command buttons labelled Requery and Reset. Note: It is easiest to copy these controls from a QBF in an existing project.
 

6. Add a label directing the user to double-click a record to open it for editing.


7. Rearrange controls appropriately.
 

8. Add some VBA code to a class module associated with your QBF form. 
Note: If you do not see Option Explicit as the 2nd line of code in newly created modules, then please see this link to properly configure your Visual Basic Editor (VBE):

Always Use Option Explicit
http://www.access.qbuilt.com/html/gem_tips.html#VBEOptions
Option Compare Database

Option Explicit

Private Function RequerySubform()

  'This is the main function that you will be calling. We will add the code later.

End Function

cmdClose

Private Sub cmdClose_Click()
On Error GoTo ProcError

      DoCmd.Close acForm, Me.Name

ExitProc:

     Exit Sub

ProcError:

     MsgBox "Error " & Err.Number & ": " & Err.Description, _

                     vbCritical, "Error in procedure cmdClose_Click..."

     Resume ExitProc

End Sub

cmdRequery

     (  Event Procedure:    =RequerySubform()

Select all controls on the main form and add the same call as above to the After Update Event. The idea is that we will run a function named RequerySubform every time a user updates a control.
cmdReset
Private Sub cmdReset_Click()

On Error GoTo ProcError

   Dim intCurrCat As Integer

   '-- First, clear the multi-select list boxes.

   For intCurrCat = 0 To Me.lboSuppliers.ListCount - 1

       Me.lboSuppliers.Selected(intCurrCat) = False

   Next intCurrCat

   For intCurrCat = 0 To Me.lboCategories.ListCount - 1

       Me.lboCategories.Selected(intCurrCat) = False

   Next intCurrCat

   '-- Next, clear all text boxes

   Me.txtProductName = Null

   Me.txtUnitPriceLow = Null

   Me.txtUnitPriceHigh = Null

   Me.txtReorderLevelLow = Null

   Me.txtReorderLevelHigh = Null

   '-- Set fraProductStatus to Active Records

   Me.fraProductStatus = 0

   '-- Clear the subform of any results
   Me.QBFSubform.Form.RecordSource = "SELECT * FROM qryQBF WHERE FALSE"

ExitProc:

     Exit Sub

ProcError:

     MsgBox "Error " & Err.Number & ": " & Err.Description, _

             vbCritical, "Error in procedure cmdClear_Click..."

     Resume ExitProc

End Sub

Test the cmdClear functionality by making selections in the listboxes, entering values into the text boxes, and selecting the Discontinued option button. 

Add a small textbox named txtHidden. Make the font of it's associated label red, with an arrow pointing to your small hidden text box. Set the visible property for the label to No. Set the height and width of the text box to 0, with transparent border. Set the backcolor, bordercolor and forecolor to the same color as your detail section. Set the Tab Stop to No.

Set the tab order for the controls on your QBF form.

Your new QBF form with subform should look something like the following:

Step 4: Add code to the RequerySubform function

1.) To help simplify things, I like to declare separate string variables for the SELECT, WHERE, OrderBy and full SQL statement. Declare a variant variable for each list box, and one for all text box controls.

Private Function RequerySubform()

Dim strSELECT As String

Dim strWHERE As String

Dim strOrderBy As String

Dim strFullSQL As String

Dim varSuppliers As Variant

Dim varCategories As Variant

Dim varTextboxes As Variant

Dim strFrame As String

End Function

2.) Add an error handler. Initialize the strSELECT variable to the SQL statement from qryQBF. Note the space at the end of each line and the VBA line continuation characters (space + underscore). Do not include the trailing semicolon.

Private Function RequerySubform()

On Error GoTo ProcError

Dim strSELECT As String

Dim strWHERE As String

Dim strOrderBy As String

Dim strFullSQL As String

Dim varSuppliers As Variant

Dim varCategories As Variant

Dim varTextboxes As Variant

Dim strFrame As String

strSELECT = _

 "SELECT ProductID, ProductName, CompanyName AS Supplier, " _

 & "CategoryName As Category, UnitPrice, ReorderLevel, Discontinued " _

 & "FROM Categories " _

 & "RIGHT JOIN (Suppliers RIGHT JOIN Products " _

 & "ON Suppliers.SupplierID = Products.SupplierID) " _

 & "ON Categories.CategoryID = Products.CategoryID "

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

          vbCritical, "Error in RequerySubform Function..."

   Resume ExitProc

   Resume

End Function

3.) Initialize the strOrderBy variable. Add the rest of the code indicated below, sprinkling in Debug.Print statements to help with debugging (you will comment these out later).

Private Function RequerySubform()

On Error GoTo ProcError

Dim strSELECT As String

Dim strWHERE As String

Dim strOrderBy As String

Dim strFullSQL As String

Dim varSuppliers As Variant

Dim varCategories As Variant

Dim varTextboxes As Variant

Dim strFrame As String

strSELECT = _

 "SELECT ProductID, ProductName, CompanyName AS Supplier, " _

 & "CategoryName As Category, UnitPrice, ReorderLevel, Discontinued " _

 & "FROM Categories " _

 & "RIGHT JOIN (Suppliers RIGHT JOIN Products " _

 & "ON Suppliers.SupplierID = Products.SupplierID) " _

 & "ON Categories.CategoryID = Products.CategoryID "

strOrderBy = " ORDER BY ProductName, CompanyName;"

'-- If AutoRequery is set to True, or the Requery button was pressed,

'-- then re-create the Where clause for the recordsource of the subform

'-- IMPORTANT NOTE: Do not attempt to set a break point on the next line of code, or

'-- to step through this line of code using the F8 key. You will receive Error 2474 if you

'-- attempt to do this.

    If Me.fraAutoRequery Or Screen.ActiveControl.Name = "cmdRequery" Then

        '-- Store all the criteria for the Where statement into variables.

        varSuppliers = IncludeSuppliers

        varCategories = IncludeCategories

        varTextboxes = IncludeTextboxes

        strFrame = IncludeStatus

        strWHERE = "WHERE " & (varSuppliers + " AND ") & (varCategories + " AND ") _

                    & (varTextboxes) & strFrame

       Debug.Print strWHERE

       '-- Trim off trailing " AND "

       If Mid$(strWHERE, Len(strWHERE) - 4, Len(strWHERE)) = " AND " Then

          strWHERE = Left$(strWHERE, Len(strWHERE) - 5)

       End If

       '-- If no criteria was chosen, make it so the subform will be blank.

       If strWHERE = "WHERE " Then

           strWHERE = "WHERE False "

       End If

       '-- Create the new SQL String and Store it to the Recordsource.

       strFullSQL = strSELECT & strWHERE & strOrderBy

       Debug.Print strFullSQL

       Me.QBFSubform.Form.RecordSource = strFullSQL

       '-- Set the requery button to black.

       Me.cmdRequery.ForeColor = 0

    Else

       '-- Set the requery button to red.

       Me.cmdRequery.ForeColor = 255

    End If

   Me.txtHidden.SetFocus

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

          vbCritical, "Error in RequerySubform Function..."

   Resume ExitProc

   Resume

End Function

4.) Add four functions referenced in the above RequerySubform function:       IncludeSuppliers, IncludeCategories, IncludeStatus and IncludeTextboxes. Examples are shown below for IncludeSuppliers, IncludeStatus and IncludeTextboxes. The IncludeCategories function is nearly identical to the IncludeSuppliers function, with minor changes required for the name of the listbox, and the field name. Sprinkle Debug.Print statements in liberally to aid debugging (you will comment these out later).

Note: We are using the SQL "IN" keyword while iterating the bound column of the list box control. In this case, the bound column is the first column in the row source of each list box, which corresponds to the numeric primary key of the associated lookup tables. This is similar to using the IN construct in the criteria row of the QBE grid. For example:


Field:
SupplierID


Table:  Products


Criteria: IN (1, 3, 9, 15)    ( Equivalent to Criteria:  1 OR 3 OR 9 OR 15

Private Function IncludeSuppliers () As Variant

On Error GoTo ProcError

Dim varCategory As Variant

Dim strTempInitial As String

Dim strTemp As String

' Set strTempInitial equal to "[TableName].[FieldName]"

' plus the SQL "IN" keyword + the opening parenthesis.

    strTempInitial = "[Products].[SupplierID] In ("

    strTemp = strTempInitial

' Process all items selected in lboSuppliers

   For Each varCategory In Me.lboSuppliers.ItemsSelected()

       strTemp = strTemp & Me.lboSuppliers.ItemData(varCategory) & ", "

       ' Debug.Print "strTemp = " & strTemp

   Next varCategory

   If strTemp <> strTempInitial = True Then

      IncludeSuppliers = FinishINClause(strTemp)

   Else

      IncludeSuppliers = Null

   End If

   ' Debug.Print "IncludeSuppliers = " & IncludeSuppliers

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

                   vbCritical, "Error in IncludeSuppliers Function..."

   Resume ExitProc

End Function

Private Function IncludeTextboxes() As Variant

On Error GoTo ProcError

'Examine txtProductName text box first:

   If Not IsNull(Me.txtProductName) Then

         IncludeTextboxes = "[ProductName] Like " & Chr(34) & Me. txtProductName _

                        & "*" & Chr(34) & " AND " '<--Adds trailing wildcard"      
   End If

'Examine entries in Unit Price text boxes next:

   If Not IsNull(Me.txtUnitPriceLow) Then   'UnitPrice Field is Number (Currency)

         IncludeTextboxes = IncludeTextboxes & "[UnitPrice] >= "& Me.txtUnitPriceLow  & " AND "

   End If

   If Not IsNull(Me.txtUnitPriceHigh) Then
         IncludeTextboxes = IncludeTextboxes & "[UnitPrice] <= "& Me.txtUnitPriceHigh  & " AND " 
   End If

'Examine entries in Reorder Level text boxes next:

   If Not IsNull(Me.txtReorderLevelLow) Then   'ReorderLevel Field is Numeric

         IncludeTextboxes = IncludeTextboxes & "[ReorderLevel] >= "& Me.txtReorderLevelLow  & " AND "

   End If

   If Not IsNull(Me.txtReorderLevelHigh) Then

         IncludeTextboxes = IncludeTextboxes & "[ReorderLevel] <= "& Me.txtReorderLevelHigh  & " AND "

   End If

'Set to Null if empty...

   If IsEmpty(IncludeTextboxes) Then

      IncludeTextboxes = Null

   End If

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in IncludeTextboxes event procedure..."

   Resume ExitProc

   Resume

End Function

Private Function IncludeStatus() As String

   If (Me.fraProductStatus) <= 0 Then   'YN Field is numeric (integer)
      IncludeStatus = "[Discontinued] = " & Me.fraProductStatus & " AND "

   Else

      IncludeStatus = "[Discontinued] <= 0 AND "

   End If

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in IncludeStatus procedure..."

   Resume ExitProc

End Function

Private Sub fraAutoRequery_AfterUpdate()

On Error GoTo ProcError

   If Me.fraAutoRequery = -1 Then

      RequerySubform

   End If

'Save user's last choice as the default

   CurrentDb.Execute "UPDATE tblFEVersion SET blnAutoRequery = " _

                    & Me.fraAutoRequery, dbFailOnError

ExitProc:

   Exit Sub

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in fraAutoRequery_AfterUpdate event procedure..."

   Resume ExitProc

End Sub

5.) Add a function, FinishINClause, that is used to finish or "cap-off" the strings built by iterating the .ItemsSelected property of the list box controls.

Private Function FinishINClause(strTemp As String) As String

On Error GoTo ProcError

' Strip off trailing comma and add ")"

   FinishINClause = Left$(strTemp, Len(strTemp) - 2) & ")"

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in FinishINClause Function..."

   Resume ExitProc

End Function

Step 5: Add code to the class module associated with the QBF Subform

Open the QBF Subform in design view. Use the lasso technique to select all controls. Add a call to the DoubleClick event:  =OpenRecordForEditing()

Add the following code to the class module associated with this form. This function uses the optional WhereCondition parameter of the DoCmd.OpenForm method to cause the frmProducts form to open to the record of interest:

Option Compare Database

Option Explicit

Function OpenRecordForEditing()

On Error GoTo ProcError

   If Not IsNull([ProductID]) Then

        DoCmd.OpenForm "frmProducts", OpenArgs:=1, DataMode:=acFormEdit, _

                            WhereCondition:="[ProductID] = " & [ProductID]

      DoCmd.OpenForm Me.Parent.Name, WindowMode:=acHidden

   End If

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in OpenRecordForEditing event procedure..."

   Resume ExitProc

End Function

Step 6: Optional: Add code to the class module associated with the Products form

The purpose of the Form_Open procedure is to display the record selector and navigation buttons only if the form is opened by itself (ie. by you as the DBA). If the form is opened by a user double-clicking a record in the subform, these controls are not shown. Normally, you would not allow users to open this form with all records.

Option Compare Database

Option Explicit

Private Sub Form_Open(Cancel As Integer)

On Error GoTo ProcError

   If Not IsNull(Me.OpenArgs) Then  ' Form was opened in filtered mode from 






' double-click on QBF subform
      Me.NavigationButtons = False

      Me.RecordSelectors = False

   Else

      Me.NavigationButtons = True

      Me.RecordSelectors = True

   End If

ExitProc:

   Exit Sub

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in Form_Open event procedure..."

   Resume ExitProc

End Sub

-----------------------------------------------------

The next page includes a procedure for a command button named cmdClose on the frmProducts form. Note that it is always a good idea to check the .Dirty property of bound forms, before using code to close the form. Otherwise, you can run into situations where your edits are not saved. For more details on this issue, see Access MVP Allen Browne's write-up, available here:

      Losing data when you close a form

      http://allenbrowne.com/bug-01.html 

Note: Do not attempt to check the Dirty property for unbound forms, because this property does not exist. You will get a run-time error.

Private Sub cmdClose_Click()

On Error GoTo ProcError

   'Save any changes first
   If Me.Dirty = True Then  'Save the record, requery QBF if it is open, and close this form.       
         Me.Dirty = False

         DisplayQBF (True)

   Else

      'Record was opened for viewing only, but user made no changes, so we can

      'avoid the overhead of requerying the QBF form.

      DisplayQBF (False)

   End If

   DoCmd.Close acForm, Me.Name

ExitProc:

   Exit Sub

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in cmdClose_Click event procedure ..."

   Resume ExitProc

End Sub

This procedure is used to requery the records displayed in the subform of the QBF, in the event that editing a record causes it to no longer be included in the recordset. Add an IsLoaded function into a stand-alone module, so that it can be called from various forms.

Private Sub DisplayQBF(blnRequery As Boolean)

On Error GoTo ProcError

'This code is used to requery the subform on the QBF, in the event

'that the QBF was used to find this record for editing.

   If IsLoaded("frmProductsQBF") Then

      If blnRequery = True Then

         [Forms]![frmProductsQBF]![QBFSubform].[Form].Requery

      End If

      DoCmd.OpenForm "frmProductsQBF"

   End If

ExitProc:

   Exit Sub

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

          vbCritical, "Error in DisplayQBF procedure..."

   Resume ExitProc

End Sub

Add this function to a new stand-alone module "basUtilityFunctions" if your database does not already include this function. The Northwind sample database does include an IsLoaded function already:

Option Compare Database

Option Explicit

Public Function IsLoaded(ByVal strFormName As String) As Boolean

On Error GoTo ProcError

' Returns True if the specified form is open in Form view or Datasheet view.

Const conObjStateClosed = 0

Const conDesignView = 0

   If SysCmd(acSysCmdGetObjectState, acForm, strFormName) <> conObjStateClosed Then

      If Forms(strFormName).CurrentView <> conDesignView Then

         IsLoaded = True

      End If

   End If

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

          vbCritical, "Error in IsLoaded event procedure..."

   Resume ExitProc

End Function

Tips

Don’t forget to compile (Debug | Compile ProjectName) frequently, when in the Visual Basic Editor.

Use Debug.Print statements to assist in debugging code as you are developing. Build a little, test a little, build some more, test some more, etc. 

You can print a SQL statement to the Immediate Window (open with <Ctrl><G>). Often times, this allows you to spot a missing space character or other anomoly in your SQL statement. You can also copy a SQL statement from the Immediate Window, and paste it into the SQL view of a new query. Often times, if you have an error in your SQL statement, you will receive a more meaningful error message versus attempting to assign the same statement as the recordsource of your subform in VBA code.

Comment out or remove or all Debug.Print statements when finished.

To include date criteria, you can use code similar to the following. I like to include Allen Browne’s Popup Calendar as well. Change the field names and the names of the text boxes on your form to match:

Private Function IncludeDateRaised() As Variant

On Error GoTo ProcError

   '-- Create the DateRaised Where portion of the SQL statement.

   '-- We are including records where the date entered is greater than or equal to the user's date entry

   If Not IsNull(Me.txtDateRaisedStart) Then

      IncludeDateRaised = "([Raised_on_dt] >= #" & Me.txtDateRaisedStart & "#)"

   End If

   If Not IsNull(Me.txtDateRaisedEnd) Then

      If Len(IncludeDateRaised) > 0 Then   ' There was an entry in the txtDateRaisedStart textbox
         'IncludeDateRaised = IncludeDateRaised & " AND ([Raised_on_dt] <= #" & Me.txtDateRaisedEnd & "#)"

         IncludeDateRaised = IncludeDateRaised & " AND ([Raised_on_dt] < #" & Me.txtDateRaisedEnd + 1 & "#)"

      Else

         'IncludeDateRaised = "([Raised_on_dt] <= #" & Me.txtDateRaisedEnd & "#)"

         IncludeDateRaised = "([Raised_on_dt] < #" & Me.txtDateRaisedEnd + 1 & "#)"

      End If

   End If

   If IsEmpty(IncludeDateRaised) Then

      IncludeDateRaised = Null

   End If

ExitProc:

   Exit Function

ProcError:

   MsgBox "Error " & Err.Number & ": " & Err.Description, _

           vbCritical, "Error in IncludeDateRaised Function..."

   Resume ExitProc

End Function
